
1. Uvod

Kakor smo že ugotovili podpira programski jezik PASCAL enostavne podatkovne strukture

CHAR (znakovni tip), 

INTEGER (celo številski tip), 

REAL (realno številski tip) in

BOOLEAN (logični podatkovni tip) 

določeni prevajalniki PASCALa podpirajo še SHORTINT, LONGINT, BYTE, WORD, SINGLE, DOUBLE EXTENDED, COMP

in sestavljene podatkovne tipe:

ARRAY (polje),

RECORD (zapis) in

SET (množica).

Vse zgoraj navedene podatkovne strukture so statične. Za spremenljivke, ki pripadajo tem podatkovnim tipom , prevajalnik vnaprej (v času prevajanja) rezervira pomnilniški prostor. Ko se program začne izvajati imajo spremenljivke natanko opredeljeno pomnilniško lokacijo. 

So nekateri programski problemi, kjer programer ne ve v času razvijanja aplikacije koliko bo velikost določenih podatkov. To običajno rešuje tako, da rezervira pomnilniški prostor za največjo količino podatkov, ki se lahko pojavi. Toda izkoriščenost pomnilniškega prostora v tem primeru ni racionalna.

Določeni programski jeziki podpirajo poleg statičnih podatkovnih struktur, tudi dinamične podatkovne strukture - KAZALCE ali reference. Med njimi spada tudi programski jezik PASCAL. Prednost kazalcev v primerjavi z statičnimi strukturami je v tem, da program uporabi samo toliko pomnilniškega prostora, kot ga dejansko aplikacija potrebuje.

"Slabost" kazalcev v primerjavi s statičnimi strukturami je v tem, da prevajalnik ne more (v času prevajanja) kontrolirati nekaterih napak, ki se lahko pojavijo med izvajanjem programa. Pri kazalcih mora programer sam skrbeti za dinamično dodeljevanje pomnilnika in sprostitev dodeljenega pomnilnika. Za programerja -začetnika je enostavneje, če uporablja statične podatkovne strukture, saj mu ni treba skrbeti za rezerviranje pomnilnika in ne za sprostitev zasedenega pomnilnika.

1.1. Definicija kazalčenga tipa in deklaracija kazalčnih spremenljivk

V PASCALu lahko deklariramo kazalec na celo število z naslednjim stavkom:

 var p:^integer;

Ta deklaracija pove, da je p kazalec, ki kaže na objekt tipa integer. Na splošno lahko deklariramo kazalec na poljubno pomnilniško podatkovno strukturo (namesto podatkovnega tipa integer napišemo drug podatkovni tip). 

Kazalci imajo praktični smisel v programiranju šele, če z njimi tvorimo posamezne strukture. Pogosto uporabljena struktura je seznam. Za primer si oglejmo definicijo podatkovnega tipa seznama celih števil. V seznam so shranjena cela števila in pripadajoči kazalci na naslednji element. Elementi seznama so zapisi, ki imajo po dve komponenti:

celo število 

kazalec, ki kaže na naslednji element v seznamu.

Naj bodo zapisi tipa vozel. Komponenti zapisa pa naj bosta stevilo in naslednji. Definicija podatkovnih tipov za ta primer bi lahko bila:

type     kazal=^vozel;


vozel= record


   stevilo: integer;



   naslednji: kazal



end;

Kot opažamo, je podatkovni tip kazal tipa kazalec na nekaj (vozel); prevajalniku ta tip še ni znan. S tem smo prekršili osnovna pravila definicije podatkovnih tipov  v PASCALu. To je edini primer, da lahko definiramo podatkovni tip na nekaj, kar bo šele sledilo v nadaljevanju definicije podatkovnega tipa. Če ne bi obstajal podatkovni tip vozel v nadaljevanju definicije podatkovnega tipa, potem bi prevajalnik javil napako.

Sedaj lahko deklariramo spremenljivko s:


var s: kaz;

1.2. Procedure za delo s kazalci

Spremenljivka s ima zaenkrat nedefinirano vrednost. Objekte seznama je potrebno zgraditi v času izvajanja programa. Standardna pascalska procedura za dodeljevanje pomnilniškega prostora določenemu kazalcu je: 

procedure new(var p:pointer);

Klic procedure

new(s);

povzroči, da se pojavi nov objekt tipa vozel, nanj pa kaže kazalec s. Kaj se dejansko zgodi? Del pomnilniškega prostora z velikostjo, kot je podatkovni tip vozel se zasede in spremenljivki s se dodeli pomnilniški naslov prve pomnilniške lokacije, ki jo ima ta zasedeni del pomnilnika. Vrednost objekt tipa vozel je s^ . Do posameznih komponent stevilo in naslednji pridemo z

s^.stevilo in s^.naslednji

Trenutno sta obe vrednosti (s^.stevilo in s^.naslednji) še nedefinirani. Priredimo želeno vrednost številu:

s^.stevilo:=5;

s^.naslednji:=nil;

Z zadnjim stavkom smo kazalcu dodelili posebno vrednost 'nil'. Ta vrednost je v pascalu predvidena za kazalce za primer, da kazalec ne kaže nikamor. Uporabljamo jo za označevanje konca strukture.

Obratna procedura, kot je new je procedura za sproščanje pomnilnika, ki ga je zasedel naslov, na katerem kaže nek kazalec.

procedure dispose(var p:pointer) 

Če bi poklicali

dispose(s)

potem sprostimo pomnilnik v velikosti tipa vozel in spremenljivka s ima nedefinirano vrednost. Sam jezik PASCAL ne govori o tem ali program po končanem izvajanju samodejno sprosti ves zasedeni pomnilniški prostor, ki je bil dodeljen dinamično s procedure new. Odvisno je od samega prevajalnika in operacijskega sistema na katerem teče program. Nekateri operacijski sistemi vnaprej določijo fiksen pomnilniški prostor programu, ki se izvaja. Pri teh operacijskih sistemih kazalec, ki je parameter procedure new dobi pomnilniški prostor iz tega fiksnega dela pomnilnika. Po končanem izvajanju programa, se sprosti ves pomnilniški prostor, ki ga je program zasedel (hkrati  s kazalčnim pomnilnikom). Drugi operacijski sistemi imajo globalni pomnilniški prostor in klic procedure new zasede globalni pomnilniški prostor, ki je namenjen vsem programom. To pomeni, da pri takih operacijskih sistemih programer mora sam "počistiti za sabo" (sprostiti vse kazalčne vrednosti s proceduro dispose, ki jih je zasedel v času izvajanja programa). Torej, dober programer počisti vse zasedene pomnilniške naslove, ki jih je zasedel s klicem procedure new.

1.3. Prirejanje

Predpostavimo, da imamo dve spremenljivki istega podatkovnega tipa, recimo:

var i,j:integer;

Nekje v programu lahko priredimo določeni spremenljivki konstantno vrednost tipa integer 

i:=1; j:=50;

Prirejanje vrednosti neki spremenljivki je lahko tudi neka druga spremenljivka istega tipa. Če bi v našem primeru zapisali:

i:=j;

potem spremenljivka i  bi dobila vrednost 50.

Kako je s prirejanjem kazalcev?

S tega stališča se kazalčne spremenljivke vedejo enako kot običajne spremenljivke. 

type     kazal=^vozel;

vozel= record

   stevilo: integer;


   naslednji: kazal


end;

var s,p: kazal;

begin


new(s);


s^.stevilo:=6;


s^.kazal:=nil;


p:=s;

end.

Predzadnji stavek prejšnjega programa priredi kazalcu p enako vrednost, kot jo ima kazalec s. 

1.4. Primerjanje kazalcev

V pogojnih stavkih je možno primerjati kazalce. V prejšnjem primeru, če bi  dodali pred end-om

if (p=s) then ...

pogoj (p=s) bi imel vrednost true.

Ni priporočljivo testirati vrednost kazalcev. Običajno ne primerjamo same kazalce (pomnilniške naslove) temveč vrednosti, ki so v lokacijah na katerih kažejo kazalci:

if (p^.stevilo= s^.stevilo) then...

2. ZGLEDI

Skupaj s tem priročnikom so še izvorna koda za vse zglede. Izvorne kode so priložene v naslednjih datotekah:

Zgled
PASCALSKI program (ime datoteke)

zgled 1
zgled1.pas

zgled 2
zgled2.pas

zgled 3 
zgled3.pas

zgledi 4 do 7
neurejsz.pas

zgledi 8 do 13
urejensz.pas

1. Sestavi program, ki prebere niz znakov v dinamično spremenljivko. Izpiši niz brez presledkov!

program Zgled1;

(*Program prebere niz znakov v dinamično spremenljivko in izpise niz brez 
presledkov*)

type kString=^string;

var 
nizZnakov:kString;


dolzinaNiza,i:integer;

begin


new(nizZnakov);


write('Vpisi niz znakov '); readln(nizZnakov^);


dolzinaNiza:=Length(nizZnakov^);


for i:=1 to dolzinaNiza do

if nizZnakov^[i]<>' ' then write(nizZnakov^[i]);


readln

end.
V programu definiramo podatkovni tip KString, to je kazalec na string. Deklariramo še spremenljivko nizZankov, ki je kazalec na string. Spremenljivko dolzinaNiza potrebujemo, da v 3. vrstici po begin stavku enkrat izračunamo dolžino stringa s funkcijo lenght(nizznakov^), drugače bi se ta dolžina računala v for zanki toliko krat, kolikor je dolžina stringa (program bi po nepotrebnem računal dolžino večkrat). i je pomožna spremenljivka v for zanki. 

S proceduro new(nizZnakov) rezerviramo pomnilniški prostor za znake, ki jih bo vpisal uporabnik v naslednji vrsti (stavek read(nizZnakov^). S funkcijo Length(nizZnakov^) priredimo spremenljivki dolzinaNiza celo število, ki je dolžina niza. V for zanki testiramo s pogojnim stavkom if ali je določen znak polja znakov različen od znaka presledek če je različen, ga izpišemo na zaslon s proceduro write(nizZnakov^[i]).
Procedura readln počaka, da uporabnik pritisne na ENTER. To smo dodali samo, da ustavimo izvajanje programa, da bi videl uporabnik rezultate izvajanja.

Program ni sprostil spominskega prostora, ki ga je zasedla spremenljivka nizZnakov. Lahko bi dodali v zadnji vrsti pred end-om Dispose(nizZnakov).

Sestavi program, ki zapiše tabelo v dinamično spremeljivko! Poskusi izpisati vrednosti tabele na zaslon!

1
2
3
4
…
n

2
3
4
5
...
1

3
4
5
6
...
2

4
5
6
7
...
3

…
...
...
...
...
...

n
1
2
3
...
n-1

program Zgled2;

(*Program zapise tabelo nxn celih stevil v dinamicno spremenljivko *)

(*tabela

1,2,3,4,5...n

2,3,4,5,6,.. 1

...

n,1,2,3,...n-1

Pri izpisu, je problem, ce je dimenzija tabele vecja od 24x24, ker

je tekstovna sirina zaslona 25x80. Zadnja vrsta je pri izpisu vedno prazna writeln

*)

const DimenzijaTabele=23;

type tabelaCelihStevil=array[0..DimenzijaTabele-1,0..DimenzijaTabele-1] of integer;

     kTabeloCelihStevil=^tabelaCelihStevil;

var KazalecNaTabelo:kTabeloCelihStevil;

    i,j:integer;

begin

    new(KazalecNaTabelo);

    for i:=0 to DimenzijaTabele-1 do
    begin

      for j:= 0 to DimenzijaTabele-1 do
      begin

        KazalecNaTabelo^[i,j]:=(i+j) mod (DimenzijaTabele) +1;

        write(KazalecNaTabelo^[i,j]:3)

      end;

      writeln

    end

    readln;

    Dispose(KazalecNaTabelo)

end.

V zgledu 2 smo izbrali dimenzijo tabele 23x23 elementov. Če želimo, da program rezervira pomnilniški prostor za večjo tabelo, spremenimo vrednost konstante DimenzijaTabele. V programu smo definirali podatkovni tip kTabeloCelihStevil, ki kaže na dvodimenzionalni tabelo celih števil. Spremenljivka KazalecNaTabelo je podatkovnega tipa kTabeloCelihStevil. Spremenljivki i in j potrebujemo v obeh for zankah.

S stavkom new(KazalecNaTabelo) rezerviramo pomnilniški prostor za tabelo celih števil z dimenzijo 23x23. V notranji for zanki priredimo vrednosti tabeli po formuli  (i+j) mod (DimenzijaTabele) +1 . V naslednjem stavku write(KazalecNaTabelo^[i,j]:3) prirejeno vrednost izpišemo na zaslon. Po koncu zanke s števcem j, prelomimo vrstico s proceduro writeln (izpis za naslednjo zanko s števcem i se nadaljuje v naslednjo vrsto). Tudi zadnja iteracija zanke i prelomi vrstico t.j. program se končuje s kurzorjem v prazni vrstici. Ker je tekstovna dimenzija zaslona 25x80, lahko izpišemo 24 vrstic, ker 25 vrstica bo prazna. Bralcu damo v razmislek, da popravi algoritem, tako, da je možno izpisati "nemoteno" 25 vrstic (vse vrstice vidne na zaslonu z dimenzijo 25x80).

Na koncu programa sprostimo ves pomnilniški prostor, ki ga je zasedla dinamična spremenljivka KazalecNaTabelo.

Analiziraj kako se spreminjajo vrednosti spremenljivk p1, p2 in p3!

program Zgled3;

type kInteger=^integer;

var 
p1,p2,p3:kInteger;

begin


new(p1);
{1}


p1^:=100;
{2}


p2:=p1;
{3}


p3:=p2;
{4}


p2^:=200;
{5}


write(p1^,p2^);
{6}

end.
V razlagi se sklicujem na komentarje, ki so v vsaki vrsti desno.

V vrsti označena s komentarjem 1 spremenljivki p1 dodelimo nov pomnilniški prostor velikosti za vpis enega integer števila. Vrednost v spominu z naslovom p1 je neopredeljena - naključna. V tem trenutku imata spremenljivki p2 in p3 nedoločeno vrednost. V vrsti 2 v pomnilniškem prostoru na katerega kaže p1 zapišemo vrednost 100. V tretji vrsti kazalec p2 dobi enako vrednost kot kazalec p1. Oba kazalca p1 in p2 kažeta na isto pomnilniško besedo v pomnilniku. Vrednost kjer kažeta je 100. V 4. vrsti tudi kazalec p3 nastavimo da kaže na isto pomnilniško besedo, kot kažeta p1 in p2. V vrsti 5 vrednost, kjer kaže kazalec p2 spremenimo na 200. V vrsti 6 izpišemo vrednost, kjer kažeta kazalca p1 in p2 . Ali se izpiše 100200? Ne. Izpiše se 200200. Zakaj? Zato ker vsi kazalci kažejo na isto pomnilniško besedo. Sprememba vrednosti pomnilnika v enem od kazalcev je "vidna" še pri drugih kazalcih.

2.1. Povezani neurejeni linearni seznam

Smisel uporabe kazalcev je, če jih uporabljamo v smiselno povezanih strukturah. Pogosto uporabljena struktura je seznam. Za zgled bomo obravnavali neurejeni in urejeni linearni seznam. Razlika med urejenim in neurejenim seznamom je v tem, da ima urejeni seznam elemente seznama razvrščene po naraščajočem ali padajočem vrstnem redu (to je relacija večji ali manjši). V našem zgledu urejenega seznama so elementi cela števila, ki so v seznamu razvrščena od najmanjšega proti največjemu (na začetku seznama je najmanjši, na koncu največji element). Vstavljanje elementov v urejeni seznam je vedno na pravo mesto v seznamu(glede relacije je večji). Naš zgled neurejenega seznama vstavlja elemente vedno na začetek seznama; ne skrbimo za urejenost seznama. Enostavnejša realizacija s kazalci je neurejeni seznam.

Operacije, ki jih uporabljamo v seznamih so vstavljanje, iskanje in brisanje elementa, izpis seznama.

Za začetek si bomo ogledali enostavnejši primer, to je  neurejeni linearni seznam.

Podatkovne strukture in globalna spremenljivka, ki jih bomo uporabljali v zgledih od 4 do 7 bodo:

type    kazal=^vozel;


vozel= record


   stevilo: integer;



   naslednji: kazal



end;

var ZacetekSeznama: kazal;

2. Napiši proceduro, ki sprosti delovni spomin, ki ga je zasedel seznam! 

procedure SprostiPomnilnik(var p:kazal);

{procedura sprosti spomin in postavi klicani kazalec na vrednost nil}

begin

  if p<>nil then
  begin

    SprostiPomnilnik(p^.naslednji);

    Dispose(p)

  end;

  p:=nil

end; { SprostiPomnilnik }

Procedura za sprostitev spomina se izvaja rekurzivno. Zadnji stavek (p:=nil) se izvede toliko krat kolikor je dolžina seznama. 

Napiši funkcijo function Nasel(element:integer;p:kazal):boolean;, ki poišče element v seznamu, ter vrne vrednost true, če je element v seznamu oziroma false, če elementa v seznamu ni! 

function NaselRekurzivno(element:integer;p:kazal):boolean;

{parameter element je iskano stevilo, p je kazalec na seznam}

begin

  if p=nil then NaselRekurzivno:=false

  else

    if p^.stevilo=element then NaselRekurzivno:=true

    else NaselRekurzivno:=NaselRekurzivno(element,p^.naslednji)

end;
Seznam preiščemo do elementa, ki ga iščemo. V najslabšem primeru (če je element, ki ga iščemo zadnji v seznamu, ali če ga ni) preiščemo cel seznam. 

Sestavi podprogram, ki vstavi element v neurejen seznam!

procedure Vstavi(element:integer);

{ procedura vstavi element v neurejeni seznam}

var q:kazal;

begin


new(q);


q^.stevilo:=element;


q^.naslednji:=ZacetekSeznama;

{3}


ZacetekSeznama:=q;


{4}

end;

Procedura Vstavi vstavi element na začetek neurejenega seznama. S stavkom q^.naslednji:=ZacetekSeznama kazalec naslednjega od q kaže na isto pomnilniško lokacijo, kot ZacetekSeznama. Zadnji stavek ZacetekSeznama:=q postavi vrednost spremenljivki ZacetekSeznama.

Kaj bi se zgodilo, če bi zamenjali med seboj vrsti označeni s komentarjem 3 in 4? Odgovor je preprost: izgubili bi seznam; v seznamu bi bil en sam element in ne bi bilo kazalca na nil. To si ne smemo privoščiti.

3. Sestavi podprogram, ki izpiše elemente seznama!

procedure Izpisi_Seznam_Kazalci;

var p: kazal;

begin

{izpis vseh elementov seznama}

  p:=zacetekSeznama;

  while p<>nil do
  begin

    write(p^.stevilo:5);

    p:=p^.naslednji

  end

end;
Na začetku procedure postavimo pomožni kazalec p na začetek seznama (ne smemo direktno delati s kazalcem zacetekSeznama, ker bi izgubili začetek seznama). Seznam obhodimo do konca (do vrednosti nil) z zanko while. S stavkom write(p^.stevilo:5) izpišemo element seznama na zaslon. Pred koncem zanke while kazalec p nastavimo na naslednji element seznama.

2.2. Povezani urejeni linearni seznam

Sedaj si oglejmo še urejeni linearni seznam.

Podatkovne strukture in globalna spremenljivka, ki jih bomo uporabljali v zgledih od 8 do 13 bodo:

type    kazal=^vozel;


vozel= record


   stevilo: integer;

          
   strazar:boolean;



   naslednji: kazal



end;

var ZacetekSeznama: kazal;

Običajno konec seznama s kazalcih je opredeljen z vrednostjo NIL, ki jo definira PASCAL. Zakaj smo vpeljali pomožno vrednost v podatkovnem tipu vozel, bo postalo jasno, po razlagi procedure za vstavljanje v urejeni seznam.  Seznam bo vedno vseboval vsaj en element tipa vozel, ki bo pri iskanju, izpisu, itn. nepomemben (ta bo vedno zadnji element seznama). Funkcija tega elementa je samo stražar pred koncem (NILom). Res, da bi lahko vstavili tudi neko globalno spremenljivko, ki bi hranila naslov zadnjega kazalca, toda zaradi različnega načina naslavljanja nekaterih procesorjev (da primerjava dveh kazalcev lahko vrne vrednost FALSE, ko gre za iste kazalce), sem se odločila za redundanco v podatkih (raje kot kontrola naslovov procesorjev). 

4. Napiši proceduro, ki inicializira seznam kazalcev in spremenljivko ZacetekSeznama postavi na kazalčno vrednost, ki kaže na začetek seznama!

procedure Init(var p:kazal);

begin

  new(p);

  p^.naslednji:=nil;

  p^.strazar:=true

end; {init}

Proceduro pokličemo iz glavnega programa s parametrom ZacetekSeznama. Ker je spremenljivka p klicana po referenci, bo ZacetekSeznama po klicu procedure kazal na en element tipa kazal, z vrednostjo strazar=true; naslednji postavimo na vrednost nil. Ker funkcija tega elementa je samo strazar na konec seznama, brezpredmetno je ali nastavimo vrednost p^.stevilo na neko vrednost ali ne.

5. Napiši proceduro, ki sprosti delovni spomin, ki ga je zasedel seznam! Kako bi izvedli celotno brisanje seznama in priprava seznam za ponovno vstavljanje elementov v seznam?

procedure SprostiPomnilnik(var p:kazal);

{procedura sprosti spomin in postavi klicani kazalec na vrednost nil}

begin

  if p<>nil then
  begin

    SprostiPomnilnik(p^.naslednji);

    Dispose(p)

  end;

  p:=nil

end; { SprostiPomnilnik }

Procedura za sprostitev spomina se izvaja rekurzivno. Zadnji stavek (p:=nil) se izvede toliko krat kolikor je dolžina seznama+1. Da bi zmanjšali število primerjanj, bi lahko testirali ali smo prvič vstopili v rekurzijo z dodatnim parametrom tiba boolean (glava procedure bi bila SprostiPomnilnik(var p:kazal; prvic:boolean) ), klic vseh rekurzivnih SprostiSpomin(p^.naslednji,false);.  Pred predzadnjo vrsto bi dodali pogojni stavek 

if prvic then p:=nil;

vendar, vprašljivo je kaj se hitreje izvede primerjanje ali prirejanje (odvisno je od procesorja ali ima hitrejše implementirano operacijo primerjanje ali prirejanje).

Brisanje in priprava seznama za ponovni vnos izvedemo iz glavnega programa s klicem 

SprostiPomnilnik(ZacetekSeznama);

Init(ZacetekSeznama);

Napiši funkcijo function Nasel(element:integer;p:kazal):boolean;, ki poišče element v seznamu, ter vrne vrednost true, če je element v seznamu oziroma false, če elementa v seznamu ni! 

function NaselRekurzivno(element:integer;p:kazal):boolean;

{parameter element je iskano stevilo, p je kazalec na seznam}

begin

  if (p^.strazar or (p^.stevilo>element)) then NaselRekurzivno:=false

  else

    if p^.stevilo=element then NaselRekurzivno:=true

    else NaselRekurzivno:=NaselRekurzivno(element,p^.naslednji)

end;
Prva rešitev je rekurzivna. Seznam preiščemo do stražarja ali do prvega elementa v seznamu, ki je večji od iskanega elementa (ker gre za urejen seznam, tukaj iskanje prekinemo). V čem je razlika, če preiskujemo urejen ali neurejen seznam. Pri neurejenem seznamu, potrebno je preiskati celoten seznam do konca (v povprečju je to počasnejše), toda enostavnejše so operacije vstavljanja elementa v neurejenem seznamu (ne potrebujemo stražarja). 

Realizacija iste naloge iterativno je naslednja:

function NaselIterativno(element:integer;p:kazal):boolean;

{parameter element je iskano stevilo, p je kazalec na seznam}

var najden:boolean;

begin

  najden:=false;

  while ((not p^.strazar) and (p^.stevilo<=element) and (not najden)) do
   if p^.stevilo=element then najden:=true

   else  p:=p^.naslednji;

  NaselIterativno:=najden

end;
Spremenljivko najden postavimo na vrednost false. Z zanko while preiskujemo seznam do konca oz. dokler ne najdemo element v seznamu, ki je enak ali večji od iskanega elementa. Če smo prišli do elementa, ki je večji od iskanega elementa (ker gre za urejen seznam) prekinemo iskanje z ugotovitvijo, da ni takega elementa v seznamu.

Sestavi podprogram, ki vstavi element v urejen seznam!

procedure Vstavi(var p:kazal; element:integer);

{ procedura vstavi element v urejeni seznam}

var q:kazal;

begin

  if ((p^.strazar) or (p^.stevilo>=element)) then
   begin

    new(q);

    q^:=p^;

    p^.stevilo:=element;

    p^.strazar:=false;

    p^.naslednji:=q;

   end

  else  Vstavi(p^.naslednji,element)

end;

V našem seznamu je vedno vsaj en element, ki ima vrednost logične vrednosti strazar=true. Rekurzivno preiščemo seznam dokler ne najdemo elementa, ki je enak ali večji od elementa, ki ga vstavljamo oz. dokler ne pridemo do stražarja na koncu seznama (če je element, ki ga vstavljamo večji od vseh elementov, ki so trenutno v seznamu). Kazalec p kaže na najdeni element ali na stražarja. S stavkom new(q), rezerviramo pomnilniški prostor. S stavkom

q^:=p^;
priredimo novi spremenljivki vse vrednosti enake, kot jih je imel p^. p^.stevilo  priredimo vrednost elemetn, p^.strazar vrednost false in p^.naslednji vrednost q.
Dejansko smo dodali element za najdenim elementom, vendar smo prepisali vrednosti iz najdenega elementa v q^.

6. Sestavi podprogram, ki izpiše elemente seznama!

procedure Izpisi_Seznam_Kazalci;

var p: kazal;

begin

{izpis vseh elementov seznama, razen strazarja}

  p:=zacetekSeznama;

  while not p^.strazar do
  begin

    write(p^.stevilo:5);

    p:=p^.naslednji

  end

end;
Na začetku procedure postavimo pomožni kazalec p na začetek seznama (ne smemo direktno delati s kazalcem zacetekSeznama, ker bi izgubili začetek seznama). Ker je v našem seznamu vedno na koncu stražar, seznam obhodimo do stražarja z zanko while. S stavkom write(p^.stevilo:5) izpišemo element seznama na zaslon. Pred koncem zanke while kazalec p nastavimo na naslednji element seznama.
7. Sestavi podprogram, ki izloči vse elemente, ki se pojavljajo v seznamu dvakrat!

procedure PobrisiDvojne(var p:kazal);

var tmpKazalec:kazal;

begin

  if not p^.strazar  then

  begin

    PobrisiDvojne(P^.naslednji);

    if NaselIterativno(p^.stevilo,p^.naslednji) then
    begin

      tmpKazalec:=p;

      p:=p^.naslednji;

      dispose(tmpKazalec)

    end

  end

end;

Tudi ta rešitev je elegantneje napisana, če je algoritem rekurziven. V proceduri uporabljamo proceduro NaselIterativno, ki smo jo obravnavali v zgledu 5.

Seznam preiskujemo od repa proti začetku. Izločanje izvedemo s stavkom p:=p^.naslednji. Spremenljivko tmpKazalec uporabljamo s ciljem, da bi sprostili pomnilnik, ki ga je zasedla izločena vsebina. 

3. Vaje za utrjevanje snovi

8. Definiraj podatkovni tip kazalec na znakovni tip (char)!

9. Deklariraj dva kazalca na cela števila. Vrednosti, kamor kažeta kazalca zamenjaj med sabo!

10. Sestavi podprogram, ki izbriše določen element iz urejenega seznama! Podatkovne strukture so enake kot v zgornjih zgledih.

11. Sestavi podprogram, ki v seznam iz zgleda vstavi 500 elementov (naravna števila od 1 do 500)!

12. Podprogram Izpisi_Seznama_Kazalci izpiše urejeni linearni seznam naraščajoče. Sestavi nov podprogram, ki naraščajoče urejeni seznam izpiše padajoče. Pomagaj si z rekurzijo!

Uporabi podprogram Vstavi iz zgleda 6 in sestavi podprogram, ki iz 100 naključno izbranih celih števil od 1 do 1000 naredi neurejeni linearni seznam. Preveri delovanje podprograma tako, da ta seznam še izpišeš!

Uporabi podprogram Vstavi iz zgleda 11 in sestavi podprogram, ki iz 100 naključno izbranih celih števil od 1 do 1000 naredi urejeni linearni seznam. Preveri delovanje podprograma tako, da ta seznam še izpišeš!


13. Sestavi preprost uporabniški vmesnik, ki združuje procedure in funkcije za delo z urejenim seznamom! Vmesnik naj omogoča uporabniku klicanje vseh procedur in funkcij iz naših zgledov.

Analiziraj, kakšni problemi se lahko pojavijo v proceduri Vstavi(var p:kazal; element:integer); če ne bi bilo stražarja?
Kaj manjka naslednji procedure (pri predpostavki, da ni stražarja- prazen seznam je nil)? Za katere primere  algoritem ne dela pravilno?

procedure Vstavi(var p:kazal; element:integer);

{ procedura vstavi element v urejeni seznam}

var q:kazal;

begin

  if ((p=nil) or (p^.stevilo>=element)) then
   begin

    new(q);

    q^:=p^;

    p^.stevilo:=element;

    p^.naslednji:=q;

   end

  else  Vstavi(p^.naslednji,element)

end;

14. Binarno drevo je definirano kot

type vozel=record stevilo:integer;

                  levi,desni:^vozel

            end;

     kaz=^vozel;

V levem poddrevesu so vedno manjše vrednosti, v desnem, pa večje vrednosti.

Napiši proceduro Vstavi(x:integer;var p:kaz); , ki vstavi določen element v binarnem drevesu!

15. Napiši funkcijo, ki poišče ali je nek element v binarnem drevesu! Funkcija vrne vrednost true, če je element v seznamu oz. false, če elementa ni v seznamu.

16. Nekateri programski jeziki ne podpirajo dela s kazalci. Sestavi podatkovno strukturo s polji, ki nadomesti naslednjo strukturo:

type    kazal=^vozel;


vozel= record


   stevilo: integer;



   naslednji: kazal



end;

var ZacetekSeznama: kazal;

Zgled naj podpira do 1000 elementov. Realiziraj podprograme New, IzpisiSeznam, VstaviElement.

